M.Albenque www.
liafa.jussieu.fr/~albenque/Maple.htm

albenque@liafa.jussieu.fr

MK1 "Calcul formel" Maple

TP5 : Résolution d'équations, dérivation, intégration
Limites, continuité

Commandes pour les caractères spéciaux
Voici les commandes pour obtenir les caractères qui ne sont pas présents sur les claviers Mac.
pour ~ : Alt n
pour { : Alt (pour [: Alt Shift (
pour } : Alt)   pour] : Alt Shift)

Et surtout, n'oubliez pas de vous (et de me) poser des questions !

1. Résoudre des équations

1.1 Résoudre une équation à une inconnue

Pour résoudre une équation à une inconnue, on utilise la commande *solve* dont la syntaxe est *solve (equation, variable).*
L'équation est définie par une égalité =, à bien différencier des affectations, qui se font par : =.

Commençons par les **équations polynomiales**. Maple résout complètement sur **C** les équations polynomiales de degré <=3 et quelques équations de degré supérieur.

```
> restart;
  solve(x=1+1/x,x);
```

$$\frac{1}{2}+\frac{\sqrt{5}}{2},\ \frac{1}{2}-\frac{\sqrt{5}}{2} \qquad (1)$$

Ici, Maple donne les solutions exactes et explicites de l'équation. Le type de résultat renvoyé par Maple est une *séquence.* On peut demander les valeurs approchées des solutions par :

```
> evalf(%);
```

$$1.618033988,\ -.6180339880 \qquad (2)$$

Malheureusement, en degré supérieur ou égal à 4, cela ne se passe pas toujours aussi bien.
Essayons de résoudre l'équation : x^7-x^6+x^2-1=0.

```
> S:=solve(x^7-x^6+x^2-1=0,x);
```

$$S := 1, \mathrm{RootOf}(\_Z^6+\_Z+1, index=1), \mathrm{RootOf}(\_Z^6+\_Z+1, index=2), \mathrm{RootOf}(\_Z^6+\_Z+1, index=3), \mathrm{RootOf}(\_Z^6+\_Z+1, index=4), \mathrm{RootOf}(\_Z^6+\_Z+1, index=5), \mathrm{RootOf}(\_Z^6+\_Z+1, index=6) \qquad (3)$$

Parfois, lorsque Maple ne peut pas trouver de solution explicite, ou lorsque celle-ci est trop compliquée pour être utilisable, les solutions sont exprimées à l'aide de la fonction *RootOf* ("racine de"), qui "représente" toutes les racines à la fois et aucune en particulier. Factorisons le polynôme de départ.

```
> factor(x^7-x^6+x^2-1);
```

$$(x-1)\,(x^6+x+1) \qquad (4)$$

Une racine évidente est 1, les autres racines sont celles de x^6+x+1 (au nombre de 6), que Maple refuse d'expliciter. Cependant, ce n'est pas complètement un échec car :
* certaines fonctions de Maple sont capables de travailler avec *RootOf,* et on peut continuer alors à "faire des calculs" :

```
> alias(alpha=RootOf(x^6+x+1));
```

$$\alpha \qquad (5)$$

```
> alpha^8;
  simplify(alpha^8);
```

$$\alpha^8 \qquad (6)$$

$$-\alpha^2\,(1+\alpha)$$

* si l'équation ne comporte aucun paramètre, on peut obtenir des valeurs numériques approchées des racines :

```
> evalf(S);
```

$$1., 0.9454023333+0.6118366938\,I, -0.1547351445+1.038380754\,I, -0.7906671888+0.3005069203\,I, -0.7906671888-0.3005069203\,I, -0.1547351445-1.038380754\,I, 0.9454023333-0.6118366938\,I \qquad (7)$$

La fonction *allvalues* permet parfois de "déterminer" toutes les racines désignées par le *RootOf :* quand c'est possible, Maple donne des expressions exactes.

```
> T:=solve(-x**4+5*x**3+x**2-1,x);
```

$$T := \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=1), \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=2), \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=3), \mathrm{RootOf}(\_Z^4-5\,\_Z^3-\_Z^2+1, index=4) \qquad (8)$$

```
> allvalues([T]):
  allvalues([T])[1];
```

(9)

$$\frac{5}{4}+\frac{\sqrt{3}\sqrt{\frac{83\left(2980+12\sqrt{60693}\right)^{(1/3)}+2\left(2980+12\sqrt{60693}\right)^{(2/3)}+104}{\left(2980+12\sqrt{60693}\right)^{(1/3)}}}}{12}-\frac{1}{12}\Bigg(\Bigg(\Bigg(\qquad (9)$$

$$498\left(2980+12\sqrt{60693}\right)^{(1/3)}$$

$$\left(\frac{83\left(2980+12\sqrt{60693}\right)^{(1/3)}+2\left(2980+12\sqrt{60693}\right)^{(2/3)}+104}{\left(2980+12\sqrt{60693}\right)^{(1/3)}}\right)^{\frac{1}{2}}-6$$

$$\left(\frac{83\left(2980+12\sqrt{60693}\right)^{(1/3)}+2\left(2980+12\sqrt{60693}\right)^{(2/3)}+104}{\left(2980+12\sqrt{60693}\right)^{(1/3)}}\right)^{\frac{1}{2}}$$

$$\left(2980+12\sqrt{60693}\right)^{(2/3)}-312$$

$$\left(\frac{83\left(2980+12\sqrt{60693}\right)^{(1/3)}+2\left(2980+12\sqrt{60693}\right)^{(2/3)}+104}{\left(2980+12\sqrt{60693}\right)^{(1/3)}}\right)^{\frac{1}{2}}+2610\sqrt{3}$$

$$\left(2980+12\sqrt{60693}\right)^{(1/3)}\Bigg)\Bigg/\Bigg(\left(2980+12\sqrt{60693}\right)^{(1/3)}$$

$$\left(\frac{83\left(2980+12\sqrt{60693}\right)^{(1/3)}+2\left(2980+12\sqrt{60693}\right)^{(2/3)}+104}{\left(2980+12\sqrt{60693}\right)^{(1/3)}}\right)^{\frac{1}{2}}\Bigg)^{\frac{1}{2}}\Bigg)\Bigg)$$

Maple peut aussi résoudre quelques **équations non polynomiales** simples.

```
> solve(exp(x)=1+x,x);
```

$$0 \qquad (10)$$

Pour les équations plus compliquées, Maple peut trouver une partie des solutions, aucune des solutions, ou meme des solutions qui sont fausses ! Voir la feuille d'exercices.

1.2 Résoudre un système d'équations

Pour résoudre un système d'équations, on utilise la commande *solve* avec la syntaxe : *solve({equations}, {variables})*
Maple retourne les solutions sous la forme {variable1=expression1, variable2=expression2,...}.

Exemple : le système {x+y+z=4, x+y-z=1, x-y-z=-3}.

```
> solve({x+y+z=4,x+y-z=1,x-y-z=-3},{x,y,z});
```

$$\left\{x=\frac{1}{2}, z=\frac{3}{2}, y=2\right\} \qquad (11)$$

Ici, Maple trouve une solution unique.

Dans certains cas, on a une famille de solutions, qui s'expriment en fonction de certaines des variables.
Exemple : le système { x-3y=2,3x-9y=6}.

```
> solve({x-3*y=2,3*x-9*y=6});
```

$$\{x=3\,y+2, y=y\} \qquad (12)$$

Maple a choisi de garder l'inconnue y comme paramètre, et a pu exprimer toutes les solutions en fonction de ce paramètre.

1.3 Résolution approchée

Lorsqu'on ne peut résoudre de manière exacte une équation (sans paramètre), on peut essayer la fonction *fsolve* pour obtenir des solutions *approchées*.

```
> fsolve(tan(sin(x))=1,x);
```

$$0.9033391108 \qquad (13)$$

On peut spécifier un intervalle sur lequel on cherche les solutions :

```
> fsolve(x^5-7*x^4+3*x^2+15,x,0..10);
```

$$1.386962979, 6.931051639 \qquad (14)$$

Par défaut, Maple cherche des solutions approchées réelles. On peut aussi lui demander des solutions approchées complexes :

```
> fsolve(x^5-7*x^4+3*x^2+15,x,complex);
```

$$-1.242272278, -0.03787117016-1.120100886\,\mathrm{I}, -0.03787117016 \qquad (15)$$
$$+1.120100886\,\mathrm{I}, 1.386962979, 6.931051639$$

2. Dérivation et intégration

2.1 Dériver une expression

On dérive une *expression* à l'aide de la commande *diff*, par la syntaxe *diff(expression_en_variable, variable).*

```
> diff(1/(1+x^2),x);
```

$$-\frac{2\,x}{\left(1+x^2\right)^2} \qquad (16)$$

Maple retourne alors une *expression* en la variable.

2.2 Dériver une fonction

On dérive une *fonction* (préalablement définie à l'aide de -> ou de unapply) par l'opérateur *D*.

```
> restart; f:=x->cos(x)+5*x^2-3;
```

$$f:=x\rightarrow\cos(x)+5\,x^2-3 \qquad (17)$$

```
> g:=D(f);
```

$$g:=x\rightarrow-\sin(x)+10\,x \qquad (18)$$

Maple retourne alors une *fonction*.

```
> g(Pi);
```

$$10\,\pi \qquad (19)$$

2.3 Calculer une primitive

On utilise la commande *int*, dont la syntaxe est *int(expression_en_variable,variable)*. Le résultat est sans constante d'intégration. C'est une expression en la variable.

```
> int(tan(x),x);
```

$$-\ln(\cos(x)) \qquad (20)$$

Parfois, Maple ne sait pas calculer exactement une primitive.

```
> int(exp(x)*cos(x)^n,x);
```

$$\int e^x \cos(x)^n \,dx \qquad (21)$$

Vérifier qu'ici, si l'on donne une valeur précise à n, Maple sait mener le calcul.
Parfois, Maple exprime les primitives en fonction de fonctions dites spéciales :

```
> int(sin(x)/x,x);
```

$$\mathrm{Si}(x) \qquad (22)$$

C'est la fonction sinus intégral (pour en savoir plus, consulter l'aide sur Si). Maple la connait, il sait par exemple en tracer le graphe :

```
> plot(Si);
```



2.4 Calculer une intégrale

Pour calculer l'intégrale de a à b de f, on utilise la commande *int* avec la syntaxe *int (expression_en_variable, variable=a..b).*

```
> restart; int(exp(x)*cos(x)^3,x=0..Pi/2);
```

$$\frac{3}{10}\, e^{\left(\frac{\pi}{2}\right)} - \frac{2}{5} \qquad (23)$$

Les bornes ne sont pas nécessairemment des constantes. Elles peuvent etre des indéterminées :

```
> int(x^2*cos(x),x=0..N*Pi);
```

$$N^2\,\pi^2 \sin(N\,\pi) - 2\sin(N\,\pi) + 2\,N\,\pi\cos(N\,\pi) \qquad (24)$$

Elles peuvent etre aussi +l'infini (noté *infinity*) ou -l'infini (noté *-infinity*).

```
> int(exp(-x^2)*cos(x),x=0..infinity);
```

$$\frac{1}{2}\sqrt{\pi}\, e^{\left(\frac{-1}{4}\right)} \qquad (25)$$

(mais vous n'aurez pas de telles intégrales dans votre cours de mathématiques du premier semestre)

3. Limites

Pour calculer des limites d'expressions en une variable, on utilise la commande *limit* avec la syntaxe *limit(expression_en_x,x=a)*. La valeur de *x* pour laquelle on cherche la limite de l'expression est *a*.

```
> restart; f:=x->(1-cos(x))/x^2 ;
```

$$f := x \rightarrow \frac{1-\cos(x)}{x^2} \qquad (26)$$

La fonction f n'est pas définie en 0, mais a une limite (finie) en 0 que Maple peut calculer :

```
> f(0);
  limit(f(x),x=0);
Error, (in f) numeric exception: division by zero
```

(27)

$$\frac{1}{2}$$

a peut valoir + l'infini et -l'infini (infinity et -infinity) :

```
> limit(ln(x),x=infinity);
```

$$\infty \qquad (28)$$

```
> limit(1/x,x=0);
```

$$\mathit{undefined} \qquad (29)$$

Ici, Maple répond 'undefined' car la fonction x->1/x n'a pas de limite en 0. Par contre, elle a une limite à gauche et une limite à droite :

```
> limit(1/x,x=0,left); limit(1/x,x=0,right);
```

$$-\infty \qquad (30)$$

$$\infty$$

Pour définir une suite dont on connaît le terme général, on procède comme pour une fonction "ordinaire" : par exemple, pour la suite u_n définie par u_n=cos(1/n) :

```
> u:=n->cos(1/n);
```

$$u := n \rightarrow \cos\left(\frac{1}{n}\right) \qquad (31)$$

On peut alors utiliser la commande *limit* pour calculer la limite :

```
> limit(u(n),n=infinity);
```

$$1 \tag{32}$$

4. Continuité d'une fonction

Maple dispose de commandes pour étudier la continuité et la discontinuité des fonctions de **R** dans **R**. La commande *iscont* permet de tester la continuité d'une expression en *x* sur un intervalle donné. Le résultat donné par Maple est un booléen : *true* (vrai) si la fonction est continue sur l'intervalle, *false* (faux) si elle ne l'est pas. La syntaxe est *iscont(expression_en_x,x=a..b)* .

```
> restart; f:=x->1/(x-1);
  iscont(f(x),x=0..2);
```

$$f := x \rightarrow \frac{1}{x-1} \tag{33}$$

false

Ici, f n'est pas continue sur l'intervalle 0..2 car elle n'est pas continue en 1. Attention, par défaut, Maple étudie la continuité sur l'intervalle *ouvert*]a,b[. Par exemple, :

```
> iscont(f(x),x=0..1);
```

$$true \tag{34}$$

car *f* n'est pas continue au point 1, mais l'est sur l'intervalle ouvert en 1]0,1[. Il est possible de lui demander de travailler sur l'intervalle *fermé* [a,b] :

```
> iscont(f(x),x=0..1,'closed');
```

$$false \tag{35}$$

La commande *discont* donne l'ensemble des points de discontinuité. Sa syntaxe est *discont (expression_en_x,x)*.

```
> discont(f(x),x);
```

$$\{1\} \tag{36}$$

```
> discont(1/sin(x),x);
```

$$\{\pi \_Z1\sim\} \tag{37}$$

(Quel est l'ensemble des points de discontinuité ? Essayez de comprendre la dernière réponse de Maple)